

Themenübersicht Oktober 2006

- Nice to know
- Zur Definition von Plastizität
in den Programmen ANSYS und Workbench
- Anschlag mit MPC 184 Slot Joint
- Spannung kontra VOLT

- Unter anderem in der nächsten Ausgabe:

Import und Export von ASCII Daten mittels APDL ...

In eigener Sache:

Die Zusendung dieser Informationen erfolgt ausschließlich auf Wunsch des Empfängers und kann jederzeit unter www.cadfem.de beendet werden.

Wenngleich die vorliegenden Informationen mit größter Sorgfalt erstellt worden sind, weisen wir darauf hin, dass die Verwendung dieser unter Ausschluss jeglicher Gewährleistung erfolgt.

Impressum: CAD-FEM GmbH
Marktplatz 2
85567 Grafing b. München

Ansprechpartner:
Marc Vidal
mvidal@cadfem.de

Nice to know

ANSYS / Workbench

● Internet Explorer 7 – ANSYS Kompatibilität

Alle aktuell verfügbaren ANSYS Produkte sind gegenwärtig nicht für den Betrieb mit Microsoft Internet Explorer Version 7 zertifiziert. Die im Januar 2007 erscheinende Version 11 der ANSYS Produkte wird diesen Browser uneingeschränkt unterstützen. Es ist zudem geplant, einen Service Pack zur Anpassung der ANSYS 10 Produkte bereitzustellen.

● Netzwerkinstallation(Share) unter ANSYS 10

Bei älteren ANSYS Versionen (Classic) war es bisher möglich eine installierte ANSYS Version für andere Anwender über das Netzwerk freizugeben ohne diese lokal neu installieren zu müssen. Bei ANSYS 10 sind jedoch einige Anpassungen notwendig.

Zuerst sollte der Netzwerkshare auf dem Client auf einen festen Laufwerksbuchstaben gemappt werden.

(z.B. Netzlaufwerk verbinden : [\\server1\Ansys](#) Inc auf z:\).

Anschließend muss auf dem Client "ans_admin.exe" von diesem Share ausgeführt werden. Nach auswählen der Option "Configure a machine" müssen die Pfade entsprechend dem Laufwerksmapping angepasst werden. Nachfolgend werden dann die Pfade der folgenden drei Dateien ebenfalls auf diesen neuen Pfad angepasst:

ANS_EXT.TBL	(Ansys Inc\100\ANSYS\lib\intel)
tlbrlist100.ans	(Ansys Inc\100\ANSYS\gui\en-us\toolbars)
menulist100.ans	(Ansys Inc\100\ANSYS\gui\en-us\UIDL)

Sobald die Pfade korrekt editiert ist ANSYS fehlerfrei startbar.

Nice to know

ANSYS / Workbench

● Lizenz Borrowing

Seit erscheinen der Version 10 können Anwender mit einem entsprechenden Eintrag im Lizenzfile dort freigegebene Lizenzen aus dem Netzwerk auschecken und separat für eine begrenzte Zeit betreiben. Diese ist dann unabhängig vom Netzwerk des Servers startbar bis diese wieder zurückgegeben wird – oder aber die maximale Verleihdauer überschritten wird.

Ausgehend der Mobilität der Hardware ist die Verwendung dieser Funktion allerdings aktuell auf WINDOWS Systeme (seitens des Clients) limitiert.

● Extremwerte von Lösungen an Komponenten bestimmen

Zur Erweiterung des Einsatzes von Workbench werden häufig numerische Ergebnisse gesucht die leicht anhand kleiner APLD Scripte (Befehlsobjekt unter Lösung) bestimmt werden können.

Das folgenden Beispiel zeigt dies anhand der Extremwertbestimmung der Spannung einer Komponente:

```
cmsel,s,komponente      ! select nodes
nsort,u,sum            ! sort USUM
*get,my_max,sort,,max  ! get maximum in parameter my_max
alls
```

Anstelle des max im *get Befehl kann alternativ auch min verwendet werden. In der Variablen my_max steht der entsprechende Wert, auch für eine Parameterstudie in Workbench direkt zur Verfügung .

Nice to know


ANSYS / Workbench

• Workbench Personalisieren

Neben der bekannten Möglichkeit den Bericht von Workbench firmenspezifisch anpassen zu können, kann auch das Logo innerhalb der grafischen Darstellung angepasst werden.

Workbench 10 bindet hierbei stets eine als BMP hinterlegte Grafik ein, die bei Bedarf vom Anwender ausgetauscht werden kann. Diese Datei (ANSYS_150.bmp) befindet sich unter:

C:\Programme\Ansys Inc\v100\AISOL\CommonFiles\Bitmaps



Zur Definition von Plastizität in den Programmen ANSYS und Workbench

Motivation:

Auch dieser Beitrag im Newsletter ist motiviert durch zahlreiche Anfragen in unserem Support zum oben genannten Thema:

Die Eingabe von multilinearer Plastizität erfolgte in ANSYS bis zur Version ANSYS 9.0 typischerweise durch die Eingabe von diskreten Datenpaaren vom Typ [Spannung / Totale Dehnung]. So ist es auch in Workbench gewesen, jedenfalls bis zur Version 9.0. An dieser Stelle soll gleich angemerkt werden, dass es für die folgenden Ausführungen belanglos ist, ob man dabei von isotroper oder kinematischer Verfestigung ausgeht.

Ab der ANSYS Version 10.0 steht mit der APDL Passage TB,PLAS,... eine veränderte Dateneingabe zur Definition der Plastizität zur Verfügung. Damit ist es möglich multilineare Plastizität durch die Eingabe von Datenpaaren vom Typ [Spannung / Plastische Dehnung] zu definieren. Und auch in Workbench stellt man ab der Version 10.0 fest, dass dieselben Datenpaare als Eingabewerte verlangt werden.

Stress vs. Plastic Strain		
	Plastic Strain mm/mm	Stress MPa
1	0,	500,
2	2,6167e-006	550,
3	8,2921e-006	600,
4	2,3958e-005	650,
5	6,3984e-005	700,
6	1,5968e-004	750,
7	3,7565e-004	800,
8	8,3905e-004	850,
9	1,7899e-003	900,
10	3,6651e-003	950,
11	7,2339e-003	1000,
12	1,3812e-002	1050,
13	2,5589e-002	1100,
14	4,6127e-002	1150,
15	8,1088e-002	1200,
16	0,1393	1250,
17	0,23429	1300,
18	0,38637	1350,
19	0,6257	1400,
20	0,99628	1450,
21	1,5615	1500,
22	2,4116	1550,
23	3,6735	1600,
24	5,5236	1650,
*		

Workbench

```

TB, PLAS, 1, , 24, MISO
TBPT, , 0., 500.
TBPT, , 2.61671e-006, 550.
TBPT, , 8.29205e-006, 600.
TBPT, , 2.3958e-005, 650.
TBPT, , 6.39843e-005, 700.
...
...
                    
```

ANSYS

Zur Definition von Plastizität in den Programmen ANSYS und Workbench

Fragestellungen:

Zu diesen neuen Eingabemöglichkeiten entstehen folgende Fragen, die wir in diesem Newsletter Beitrag beantworten wollen:

1. ***Was ist hinsichtlich der Eingabe des E-Moduls zu beachten?***

Es ist hinreichend bekannt, dass bei Plastizität die Kurve im Fließbereich nicht steiler sein darf, als durch die Steigung des E-Moduls im Diagramm [Spannung / Totale Dehnung] vorgegeben.

2. ***Wie sind die Daten [Spannungen / Plastische Dehnungen] zu ermitteln ?***

Typischerweise findet man in Tabellenwerken Kurven, in denen die Wertepaare [Spannung / Totale Dehnung] eingetragen sind.

3. ***Was macht Workbench mit Daten aus der Version 9.0 in der Version 10.0?***


Zur Definition von Plastizität in den Programmen ANSYS und Workbench

Antworten und Lösungen:


1. Was ist hinsichtlich der Eingabe des E-Moduls zu beachten?

Hierzu diene der folgende Test, der interessante Ergebnisse liefert:
Man nehme ein lineares Element, zum Beispiel das ANSYS Element SOLID185 und bringe einen einaxialen Spannungszustand auf. Die Verschiebungen an der Einspannstelle sind nur in Längsrichtung blockiert, es stellt sich also ein dreidimensionaler Verzerrungszustand ein.

In Längsrichtung stellt man Spannungs-Dehnungsdiagramme dar. Das geht typischerweise gut im Time-History-Postprocessor /POST26 von ANSYS.
Der Versuch wird verschiebungsgesteuert durchgeführt:



Hinterlegt sind die auf der ersten Seite bereitgestellten Daten [Spannung / Plastische Dehnung]. Wir untersuchen zwei E-Moduli:



Zur Definition von Plastizität in den Programmen ANSYS und Workbench

Antworten und Lösungen:

1. Was ist hinsichtlich der Eingabe des E-Moduls zu beachten?





Diagramme aus /POST26; [Spannung / Totale Dehnung]





Diagramme aus /POST26; [Spannung / Plastische Dehnung]





Diagramme aus /POST26; [Spannung / Elastische Dehnung]

Zur Definition von Plastizität in den Programmen ANSYS und Workbench

Antworten und Lösungen:

1. *Was ist hinsichtlich der Eingabe des E-Moduls zu beachten?*

Die Antwort auf obige Frage ist damit klar: Gar nichts !!!

Der Hintergrund ist auf der vorherigen Seite dargestellt worden.

Alle Bilder in der linken Spalte stammen aus der Berechnung mit $E=200000$ MPa, in der rechten Spalte stehen die Ergebnisse der Berechnung mit $E=100000$ MPa. Die Diagramme sind stets in derselben Skalierung gezeigt, die aufgebrachte Verschiebung war bei beiden Versuchen dieselbe.

Es zeigt sich, wie erwartet, bei den Diagrammen in der ersten Zeile, dass das Material auf der rechten Seite weicher ist als auf der linken. Der E-Modul kann als Steigung im elastischen Bereich angesehen werden. Dargestellt wurde hier [Spannung / Totale Dehnung].

Die Diagramme in der zweiten Zeile zeigen [Spannungen / Plastische Dehnungen] und diese entsprechen für beide Berechnungen der Eingabe. Die Wirkung des E-Moduls kann hier nicht erkannt werden.

Die Diagramme in der dritten Zeile zeigen [Spannungen / Elastische Dehnungen] und hier erkennt man dass sich nach Überschreiten der Fließspannung die elastischen Dehnungen auch noch fortentwickeln und zwar genau mit der Steigung, die durch den E-Modul vorgegeben war.

Das ist letzten Endes auch der Grund, warum man beim Arbeiten mit diesen Plastizitätsgesetzen völlig unabhängig bei der Wahl des E-Moduls für den numerischen Erfolg der Berechnung ist. Natürlich muss man mit Sorgfalt die Eingabedaten [Spannung / Plastische Dehnungen] ermitteln, womit wir uns hier ebenfalls befassen werden.

Interessant - aber nachzuvollziehen - ist, dass sich die plastischen Verzerrungen in der rechten Spalte nicht so weit entwickeln wie links. Das aber liegt nur an der schwachen Anfangssteigung, am kleineren E-Modul also, denn damit ist vorgegeben, wie stark der Einfluss der Plastizität später sein wird.

Bleibt also festzuhalten:

Wenn man mit der Eingabe [Spannungen / Plastische Dehnungen] arbeitet, spielt der E-Modul für den numerischen Erfolg keine Rolle mehr. Allerdings muss man korrekte Daten [Spannungen / Plastische Dehnungen] eingeben.

Zur Definition von Plastizität in den Programmen ANSYS und Workbench

Antworten und Lösungen:

2. Wie sind die Daten [Spannungen / Plastische Dehnungen] zu ermitteln?

Die Antwort auf diese Frage ist mit dem dargestellten Sachverhalt einfach:

- Man nehme ein Diagramm [Spannungen / Totale Dehnungen] und greife einzelne diskrete Wertepaare ab. Am bestem man schreibt die Daten in EXCEL ein.
- Für jedes diskrete Datenpaar rechnet man

$$e_{pl} = e_{tot} - e_{el} = e_{tot} - \frac{S}{E}$$

3. Was macht Workbench mit den Daten aus der Version 9.0 in der Version 10.0?

Zur Antwort diene wiederum ein einfacher Test mit Fließspannung 200 MPa und einem E-Modul von 200000 MPa. Ergebnis: Die Daten werden richtig umgerechnet.

Structural Steel Plastic		Structural Steel Plastic	
Add/Remove Properties		Add/Remove Properties	
Young's Modulus	2,e+005 MPa	Young's Modulus	2,e+005 MPa

Stress vs. Strain			Stress vs. Plastic Strain		
	Strain	Stress MPa		Plastic Strain mm/mm	Stress MPa
1	0,	0,	1	0,	200,
2	1,e-003	200,	2	1,1e-002	215,96
3	1,208e-002	215,96	3	2,5252e-002	229,78
4	2,6401e-002	229,78	4	4,8785e-002	245,95
5	5,0015e-002	245,95	5	7,391e-002	258,16
6	7,5201e-002	258,16	6	0,10487	265,83
7	0,1062	265,83	7	0,14125	269,34
8	0,1426	269,34	8	0,18664	272,29
9	0,188	272,29	9	0,25061	277,03
10	0,252	277,03	*		

Workbench 9.0

Workbench 10.0

Nehmen wir die Zeile mit Spannung von 269,34 MPa und rechnen:

$$e_{pl} = e_{tot} - e_{el} = e_{tot} - \frac{S}{E} = 0,1426 - \frac{269,34}{200000} = 0,14125$$

Zur Definition von Plastizität in den Programmen ANSYS und Workbench

Schlussbemerkung und Ausblick:

Geometrisch lineare und nichtlineare Berechnungen

Wir wollen nicht verschweigen, dass wir in diesem Beitrag immer eine geometrisch lineare Berechnung angenommen haben (ANSYS Kommando: NLGEOM,OFF). In Workbench wären die „Großen Verformungen“ auf „Aus“ zu setzen.

Wenn man in Wirklichkeit große Verzerrungen erwartet, ist eine geometrisch nichtlineare Berechnung durchzuführen (ANSYS Kommando: NLGEOM,ON). In Workbench wären dann die „Großen Verformungen“ auf „An“ zu schalten.

In dem Zahlenbeispiel auf der vorherigen Seite liegen mit etwa 14% plastischen Verzerrungen sicher schon große Verzerrungen vor. Möchte man dann geometrisch nichtlinear rechnen, erwartet das Programm Datenpaare in Form von [Wahren Spannungen / Logarithmischen Dehnungen].

Insofern wären die ermittelten Paare [Ingenieurs-Spannungen / Ingenieurs-Dehnungen], wie wir sie hier immer angenommen haben, noch zu transformieren. Die Formeln zur Umrechnung sind bekannt (CADFEM-Support, Nichtlinearitäten-Seminar).

Es ist schwierig einen eindeutigen Zahlenwert anzugeben, ab dem man eine Verzerrung als „groß“ bezeichnet. Es kommt immer auch auf das angegebene Material an. Am besten sieht man den Unterschied, wenn man sich beide Kurven [Wahre Spannungen / Logarithmische Dehnungen] auf der einen Seite und [Ingenieurs-Spannungen / Ingenieurs-Dehnungen] auf der anderen Seite einmal ermittelt und miteinander vergleicht.

Anschlag mit MPC 184 Slot Joint




Problemstellung:

Ein Bauteil soll sich in eine Richtung um einen bestimmten Betrag frei bewegen können und dann gegen einen Anschlag fahren. Ein klassisches Problem für Kontaktelemente? Nicht immer! In manchen Fällen bietet es sich an, dieses Problem mit Slot-Joint Elementen abzubilden.

Lösung:

Das Slot _Joint Element wird zwischen jeweils einem Knoten des Bauteils und des Anschlags eingefügt. Die freie Bewegungsrichtung kann durch ein Koordinatensystem, das dem Element zugeordnet wird, angegeben werden. Über das Kommando SECSTOP kann dann der Bewegungsraum angegeben werden.

Besonders zu beachten ist bei diesem Element, dass keine Randbedingungen an den Knoten definiert werden dürfen. Außerdem muß für die Berechnung große Verformung verwendet werden. Warum dies zwingend notwendig ist, ist z.Zt. noch nicht klar, wird aber momentan mit ANSYS Inc. geklärt.



Anschlag mit MPC 184 Slot Joint

Beispiel:

```

/prep7
et,1,181
et,2,184,8      !Slot-Joint
r,1,1
mp,ex,1,210000
mp,prxy,1,0.3

/com, Steifigkeit zur Verhinderung einer
/com, Starrkörperbewegung
TB,JOINT,99,1,2,STIF
TBTEMP,0
TBDATA,,2

/com, Koord.-system des Slot-Joint Elements
local,11,,12
sectype,2,slot,u_x
secdata,1
secjoint,,11    !Koord.-system zuweisen
secstop,ux,-3,2 !Stop bzgl. Koord.-system

csys,0
block,0,10,0,10,0,5
block,12,14,0,2,0,2
vdel,all
ksel,s,loc,z,2,5
lslk,,1
asll,,1
adel,all,,,1
alls
esize,2
amesh,all
    
```

```

/com, Slot-Joint Element erzeugen
nselect,s,loc,x,10,12
nselect,r,loc,y,0
nselect,r,loc,z,0
*get,n_min,node,,num,min
*get,n_max,node,,num,max
type,2
secn,2
mat,99
e,n_max,n_min

nselect,s,loc,z,0
d,all,uz
nselect,r,loc,y,0
d,all,uy
nselect,s,loc,z,0
nselect,r,loc,x,12,14
d,all,all

/com,Löschen der Randbdg.am Slot-Joint Element
nselect,s,,,n_min
nselect,a,,,n_max
ddel,all,all

nselect,s,loc,x,0
f,all,fx,1    !Belastung auf das Bauteil
alls

/solu
nlgeom,on
solve
/post1
plns,u,x
    
```

Spannung kontra VOLT

Motivation:

Spätestens zu Beginn der ersten dynamischen elektromagnetischen Analyse steht der Anwender vor der Fragestellung der korrekten Wahl des Ansatzes bei der Bearbeitung räumlicher Aufgaben.

Neben der korrekten Beschreibung des magnetischen Potentials an permeablen Grenzflächen (hierauf soll hier nicht näher eingegangen werden) steht vor allem die Fragestellung der Behandlung des elektrisch skalaren Potentials im Vordergrund. Dies ist insbesondere für die elektrische Speisung von Spulen und die korrekte Bestimmung von Auswertegrößen von Bedeutung.

Der folgende Artikel demonstriert die Verwendung des Freiheitsgrades VOLT zu Impedanzbestimmung einer Testanordnung für beide Elementformulierungen.

Umsetzung:

Der Aufgabenstellung zugrunde liegen die Maxwell'schen Gleichungen (unter Vernachlässigung des Verschiebungsstromes).

$$\begin{aligned} \operatorname{rot} \mathbf{E} &= - \frac{\partial \mathbf{B}}{\partial t} \\ \operatorname{rot} \mathbf{H} &= \mathbf{J} \\ \operatorname{div} \mathbf{B} &= 0 \end{aligned}$$

Durch Einführung eines magnetischen Vektorpotentials \mathbf{A} und eines skalaren elektrischen Potentials der folgenden Form sind die Gleichungen 1 und 3 erfüllt:

$$\begin{aligned} \operatorname{rot} \mathbf{A} &= \mathbf{B} \\ \mathbf{E} &= - \frac{\partial \mathbf{A}}{\partial t} - \operatorname{grad} V \end{aligned}$$

Unter Berücksichtigung der Materialbeziehung und der Divergenzfreiheit der Stromdichte wird unter Verwendung der Lorenz Eichung eine eindeutige Lösung der Vektorpotenzialaufgabe möglich.

Die in dieser Form beschriebene Verwendung des elektrischen Freiheitsgrades VOLT verwendet ANSYS zur Lösung dynamischer Feldaufgaben unter der Bezeichnung „Classical Formulation“. Aus Symmetriegründen wird dabei anstelle der elektrischen Spannung – deren Integral zugrunde gelegt:

$$V = \int U dt$$

Spannung kontra VOLT

Umsetzung (Fortsetzung):

Wie in der Dokumentation ersichtlich, verwendet ANSYS die Bezeichnung VOLT für den elektrischen Freiheitsgrad elektromagnetischer Berechnungen für verschiedene Formulierungen. Neben der classical Formulation für dynamische Berechnungen – auch als solenoidal Formulation zur Abbildung von Quelltermen ohne Rückwirkung (eher für statische Berechnungen)!

Um entsprechend Aufgaben, wie die Bestimmung der Impedanz von Leiteranordnungen korrekt durchführen zu können, muss auf die classical Formulation zurückgegriffen werden.

Beispiel:

Eine entsprechend häufig vorkommende Fragestellung soll nunmehr beispielhaft unter Verwendung der classical Formulation aufgezeigt werden.

Es sei die Impedanz einer Doppelleitung in Abhängigkeit der Frequenz zu bestimmen (wobei hierbei eine räumliche Abbildung in der FE verwendet wird – um die korrekte Anwendung zu demonstrieren).

Die Modellierung erfolgt damit in gewohnter Weise – allein der Leiter wird durch Setzen der Keyoption zusätzlich mit dem elektrischen Freiheitsgrad VOLT (classical Formulation) versorgt. Neben den magnetischen Randbedingungen (Fluss parallel) werden an den Enden des abgebildeten Segmentes der Doppelleitung elektrische Randbedingungen vorgegeben, die eine Speisung mit einem Einheitsstrom und die Bestimmung des sich einstellenden elektrischen Potentials ermöglichen.

Nach der durchgeführten harmonischen Analyse werden dann die Strom und Spannungswerte verwendet um die Impedanz abzuleiten.

Da jedoch Anstelle der Spannung bei der Modellierung die classical Formulation angewendet werden musste – muß folglich die Ableitung des bestimmten elektrischen Potentials verwendet werden (Aufgrund der reinen Sinus-Funktion die Multiplikation mit $-\Omega$ hier).

Auf der folgenden Seite finden Sie den zugrunde liegenden Input, der zudem den Vergleich der Elemente SOLID117 und SOLID97 für solche Anwendungen ermöglicht.

Spannung kontra VOLT

Beispiel:

```

r=10e-3
lang=1e-3
nstep=20
f_beg=1e-6
f_end=1e3
nl=5
fuer Luft
a=10*r
rho=1.78e-8

cet=2

ndf=.75

*dim, resu, table, nstep, 4

/prep7
cyl4, a/2, 0, 0, 0, r, 360, lang
cyl4, 0, 0, 0, -90, nl*a, 90, lang
vovla, 1, 2
numm, kp

*if, cet, eq, 1, then
et, 1, 117,
et, 2, 117, 1
*else
et, 1, 97
et, 2, 97, 1
*endif
*if, cet, eq, 1, then
et, 3, 200, 7
*else
et, 3, 200, 6
*endif

esiz, ndf*r/6
type, 3
ames, 1
mat, 2
type, 2
esiz, , 1
vsweep, 1

esiz, ndf*2*r
lesi, 13, ndf*8*r
lesi, 11, ndf*3*r, , 5
lesi, 12, ndf*3*r, , 1/5
type, 3
ames, 10
mshcopy, area, 10, 11, , , lang
mat, 1
type, 1
vsweep, 3

acle, all

mp, murx, 1, 1
mp, murx, 2, 1
mp, rsvx, 2, rho

*if, cet, eq, 1, then ! 117
asel, s, ext
nsla, s, 1
nsle, r, mid
d, all, az
alls

*else ! 97

cpcy, ax, , 0, , , lang
cpcy, ay, , 0, , , lang
cpcy, az, , 0, , , lang
nsel, s, loc, x, 0
csys, 1
nsel, a, loc, x, nl*a
d, all, all
cpdel, all, , , any
csys
alls

*endif

asel, s, area, , 1
nsla, s, 1
nsle, r, corner
cp, 1, volt, all
*get, n1, node, , num, min
f, n1, amps, 1

asel, s, area, , 2
nsla, s, 1
nsle, r, corner
d, all, volt, 0
alls

/solu
antyp, harm
harf, f_beg, f_end
nsub, nstep
outres, all, all
kbc, 1
eqsl, sparse
alls
solve
    
```


Spannung kontra VOLT

Beispiel:

```

/post1

*do,jj,1,nstep,1

! Bestimme R und L
! Realteil
set,,jj,1,0
volt1_r=volt(n1)
strom1_r=1
alls

set,,jj,1,1
volt1_i=volt(n1)
strom1_i=0
alls

*get,frq,active,,set,freq
*afun,rad
pi=acos(-1)
omega=2*pi*frq

sp1_r=(-omega)*(Volt1_i) ! ACTUAL (NOT TIME INTEGRATED) COIL VLTG
sp1_i=(omega)*(Volt1_r) ! ACTUAL (NOT TIME INTEGRATED) COIL VLTG

! R_AC
r_ac=(sp1_r*strom1_r+sp1_i*strom1_i)/(strom1_r**2+strom1_i**2)

! Reactance
rea1=(sp1_i*strom1_r-sp1_r*strom1_i)/(strom1_r**2+strom1_i**2)

! Inductance (self)
l1=rea1/(omega)

! R_DC
ar_c=pi*r**2
r_dc=rho*lang/(ar_c)

! Summary
resu(jj,1)=r_ac
resu(jj,2)=rea1
resu(jj,3)=l1
resu(jj,4)=r_dc
resu(jj,0)=frq

*enddo

! PLOT THIS
/GCOLUMN,1,R_AC
/GCOLUMN,2,L
/axlab,x,Freq [HZ]
/axlab,y,L [H]
/grtyp,2
*vplot,resu(1,0),resu(1,1),3
    
```

